Synthesis of Constrained Controller in Computer Algebra System

Pavol Bisták, Peter Ťapák, Mikuláš Huba
Slovak University of Technology in Bratislava, Fac. Electr. Eng. & Info. Technology
Ilkovičova 3, 812 19 Bratislava, Slovak Republic
fax:+421265429521 email:peter.tapak@stuba.sk http: www.kar.elf.stuba.sk

Abstract
This paper discusses the design of constrained controllers based on the modes decomposition, with respect to the symbolic solutions. A triple integrator plant has been used to illustrate the design. The paper is focused on the synthesis in a computer algebra system MAPLE, which is shown "step by step" in the paper. The designed control can be applied to a broader class of systems.
Keywords: pole assignment control, time optimal control, constraints, nonlinear, third order system.

1 Introduction

The importance of the control signal constraints is generally well known. Some attempts to treat its influence can already be found in the beginning of control. Feldbaum (1966) mentions idea of two Russian engineers to improve the steel rolling mills control by a quadratic velocity feedback - an idea from 1935 which has later been rigorously elucidated by the theory of the relay minimum time systems in 50-ties. As one of the main general results of this early period, the Feldbaum's theorem about n-intervals of the optimal continuous time control (1949) should be mentioned précised later by conclusions of the well known Pontrjagin's minimum (maximum) principle around 1956. Occurrence of different parasitic phenomena typical for the minimum time control (like the relay chattering, or oscillation in the neighbourhood of the demanded states) has already in the "golden" era of the minimum time control at the end of 50-ties led to the demand on smooth solutions and quiet steady states brought by the pole-assignment control. However, also today, after more than 70 years of intensive research in the control area, the problem of the constrained control is still not sufficiently solved, even in the case of the simplest 1st order and 2nd order plants! Of course, there exists a huge amount of literature devoted to this problem. In the last decade, many textbooks appeared treating the constrained control problem by multiple approaches (see e.g. Blanchini and Miani, 1997 and several other contributions to the book by Tarbouriech and Hennet; Borelli, 2003; Goodwin, Seron and De Doná, 2005; Glattfelder and Schaufelberger, 2003; Hu and Lin, 2001; Liu and Michel, 1994; Perez, 2005; Saberi, Stoorvogel and Sannuti 1999-2000). This variety of approaches can be understood as a result of different requirements met in practice. Despite of this, it is interesting to note that majority of the approaches try to use new approaches (as e.g. the constrained predictive control or different gain scheduling approaches based e.g. on ellipsoid techniques, etc.). In difference to these new approaches, the presented paper tends to keep continuity with the development in the minimum time and the linear pole assignment area. Under slogan by F. Kafka *"to believe in progress means not to believe the progress already was..."* we try to expand the experience gained by the minimum time control and to eliminate its shortness by using the principles of the pole assignment control.

The constrained pole assignment control gives hybrid solutions with dynamics ranging from the relay minimum time systems to the linear pole assignment ones. While the general theory of the optimal control is (theoretically) able to treat system of any order, it is to note that the practical applications of the minimum time control (see e.g. Athans and Falb, 1966; Feldbaum, 1966; Pavlov, 1966) deal mostly with the 2^{nd} and the 3^{rd} order systems. Already the 3^{rd} order relay minimum time systems were that time considered to be reasonably complex both from the point of view of the design and its implementation. This is the main reason, why the higher order problems were mostly solved only offline. By modifying the optimal control sequences from the rectangular ones to softer ones by restricting the speed of the transients by the closed loop poles, the complexity of the corresponding transients and that of their description increases. It is already to see by dealing with the 2^{nd} order processes (see e.g. Huba, 2003, or 2005). On the other hand, in difference to the 60-ties years of the previous century, today we have at disposal powerful computer algebra tools that reasonably extend the scope of solvable tasks and equip a researcher in the development process. Furthermore, result of this process can be implemented by using powerful computer tools. The aim of this paper is to show how such design steps can be supported by the MAPLE V software. The paper is based on the theoretical treatment given by Bistak et al. (2006).

2 Synthesis of Algorithm in MAPLE for Linear Segment of Reference Surface

At first we clear memory and read user packages

```
>restart:with(linalg):with(plots):with(plottools):
```

We assign the closed loop poles

```
>alfa:=array(1..3,[alpha1,alpha2,alpha3]);
```

$$alfa := [\alpha 1, \alpha 2, \alpha 3]$$

We assign the system matrix and the input vector

```
>A:=matrix(3,3,[0,1,0,0,0,1,0,0,0]);b:=
matrix(3,1,[0,0,1]);
```

$$A := \begin{bmatrix} 0 & 1 & 0 \\ 0 & 0 & 1 \\ 0 & 0 & 0 \end{bmatrix} \quad b := \begin{bmatrix} 0 \\ 0 \\ 1 \end{bmatrix}$$

We define the identity matrix with size 3×3 which is used later.

```
>II:=matrix(3,3,[1,0,0,0,1,0,0,0,1]);
```

$$II := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

We assign the control signal constraints as well.

```
> U:=array(1..2);
```

$$U := \text{array}(1\,..\,2,[\ ])$$

Then the eigenvectors $\mathbf{v}_{1,2,3}$ can be computed using the matrix expression evaluation (`evalm`) according to (6)

```
>for i from 1 to 3 do
>v[i]:=evalm(inverse(alfa[i]*II-A)&*b):
>od: print(v);
```

$$\left[\begin{bmatrix} \frac{1}{\alpha 1^3} \\ \frac{1}{\alpha 1^2} \\ \frac{1}{\alpha 1} \end{bmatrix}, \begin{bmatrix} \frac{1}{\alpha 2^3} \\ \frac{1}{\alpha 2^2} \\ \frac{1}{\alpha 2} \end{bmatrix}, \begin{bmatrix} \frac{1}{\alpha 3^3} \\ \frac{1}{\alpha 3^2} \\ \frac{1}{\alpha 3} \end{bmatrix}\right]$$

Non-linear part of the 1st subsystem is created by points $\mathbf{x}_1$ as the result of backward integration of (4) on the interval $t \in \langle 0 \quad t_1 \rangle$ using $u = q_1 = U_j$ and starting from the point $\mathbf{v}_1 U_j$. The state transition matrix $e^{-\mathbf{A}t}$ and the vector $\mathbf{b}(-t)$ of the 1st sub-system are

```
>A1:=exponential(A,-t1);
b1:=map(int,(evalm(exponential(A,tau)&*b),tau=0..-t1));
```

$$A1 := \begin{bmatrix} 1 & -t1 & \frac{t1^2}{2} \\ 0 & 1 & -t1 \\ 0 & 0 & 1 \end{bmatrix} \qquad b1 := \begin{bmatrix} -\frac{t1^3}{6} \\ \frac{t1^2}{2} \\ -t1 \end{bmatrix}$$

Reference curve can be expressed as

$$\mathbf{x}_1(t_1)=\begin{bmatrix}1 & -t_1 & \frac{t_1^{\,2}}{2}\\ 0 & 1 & -t_1\\ 0 & 0 & 1\end{bmatrix} q_1\mathbf{v}_1+\begin{bmatrix}\frac{-t_1^{\,3}}{6}\\ \frac{t_1^{\,2}}{2}\\ -t_1\end{bmatrix}U_j \qquad (1)$$

i.e.

```
> x1:=evalm(A1&*v[1]*q1+U[1]*b1);
```

$$x1 := \begin{bmatrix} q1\left(\frac{1}{\alpha 1^3}-\frac{t1}{\alpha 1^2}+\frac{t1^2}{2\,\alpha 1}\right)-\frac{1}{6}U_1\,t1^3 \\ q1\left(\frac{1}{\alpha 1^2}-\frac{t1}{\alpha 1}\right)+\frac{1}{2}U_1\,t1^2 \\ \frac{q1}{\alpha 1}-U_1\,t1 \end{bmatrix}$$

The 2^nd^ subsystem can be computed in similar way, the non-linear part of the 2^nd^ subsystem is created by points $\mathbf{x}_2$ as the result of backward integration of (4) on the interval $t\in\langle 0 \;\; t_2\rangle$ using $u=q_2=U_{3-j}-U_j$ and starting from the point $\mathbf{v}_2(U_{3-j}-U_j)$, so the state transition matrix $e^{-\mathbf{A}t}$ and the vector $\mathbf{b}(-t)$ of the 2^nd^ sub-system are

```
>A2:=exponential(A,-t2);
b2:=map(int,(evalm(exponential(A,tau)&*b),tau=0..-t2));
```

$$A2 := \begin{bmatrix}1 & -t2 & \frac{t2^2}{2}\\ 0 & 1 & -t2\\ 0 & 0 & 1\end{bmatrix} \qquad b2 := \begin{bmatrix}-\frac{t2^3}{6}\\ \frac{t2^2}{2}\\ -t2\end{bmatrix}$$

Using the 1^st^ subsystem as the initial state in the 2^nd^ subsystem one gets the generalized representation of the reference surface

$$\mathbf{x}=\mathbf{x}_2((U_{3-j}-q_1),t_2)+\mathbf{x}_1(q_1,t_1)=\begin{bmatrix}1 & -t_2 & \frac{t_2^{\,2}}{2}\\ 0 & 1 & -t_2\\ 0 & 0 & 1\end{bmatrix}(\mathbf{x}_1(q_1,t_1)+q_2\mathbf{v}_2)+\begin{bmatrix}\frac{-t_2^{\,3}}{6}\\ \frac{t_2^{\,2}}{2}\\ -t_2\end{bmatrix}U_{3-j} \qquad (2)$$

```
>x12:=evalm(A2&*(x1 + q2*v[2]) + U[2]*b2    ):
```

$$x12 := \left[q1 \left(\frac{1}{\alpha 1^3} - \frac{t1}{\alpha 1^2} + \frac{t1^2}{2\,\alpha 1} \right) - \frac{1}{6} U_1\, t1^3 + \frac{q2}{\alpha 2^3} - t2 \left(q1 \left(\frac{1}{\alpha 1^2} - \frac{t1}{\alpha 1} \right) + \frac{1}{2} U_1\, t1^2 + \frac{q2}{\alpha 2^2} \right) + \frac{1}{2} t2^2 \left(\frac{q1}{\alpha 1} - U_1\, t1 + \frac{q2}{\alpha 2} \right) - \frac{1}{6} U_2\, t2^3 \right]$$

$$\left[q1 \left(\frac{1}{\alpha 1^2} - \frac{t1}{\alpha 1} \right) + \frac{1}{2} U_1\, t1^2 + \frac{q2}{\alpha 2^2} - t2 \left(\frac{q1}{\alpha 1} - U_1\, t1 + \frac{q2}{\alpha 2} \right) + \frac{1}{2} U_2\, t2^2 \right]$$

$$\left[\frac{q1}{\alpha 1} - U_1\, t1 + \frac{q2}{\alpha 2} - U_2\, t2 \right]$$

First it is necessary to solve equation
$\mathbf{x} = \mathbf{x}_3(q_3, t_3) + \mathbf{x}_2(q_2, t_2) + \mathbf{x}_1(q_1, t_1) =$

$$= \begin{bmatrix} 1 & -t_3 & \frac{t_3^{\ 2}}{2} \\ 0 & 1 & -t_3 \\ 0 & 0 & 1 \end{bmatrix} (\mathbf{x}_1(q_1, t_1) + \mathbf{x}_2(q_2, t_2) + q_3 \mathbf{v}_3) + \begin{bmatrix} \frac{-t_3^{\ 3}}{6} \\ \frac{t_3^{\ 3}}{2} \\ -t_3 \end{bmatrix} U_j \qquad (3)$$

using $t_1 = 0, t_2 = 0, t_3 = 0$ to get parameters q_1, q_2, q_3. However it is the same as to find the point of intersection of surface (2) with the line defined by vector $\mathbf{v}_3$ and the representative point. So the parametric representation of the surface mentioned above is

```
>surf_eqx:= subs({t1=0,t2=0},x12[1,1]);
```

$$surf\_eqx := \frac{q1}{\alpha 1^3} + \frac{q2}{\alpha 2^3}$$

```
>surf_eqy:= subs({t1=0,t2=0},x12[2,1]);
```

$$surf\_eqy := \frac{q1}{\alpha 1^2} + \frac{q2}{\alpha 2^2}$$

```
>surf_eqz:= subs({t1=0,t2=0},x12[3,1]);
```

$$surf\_eqz := \frac{q1}{\alpha 1} + \frac{q2}{\alpha 2}$$

The line can be expressed as

```
>line_eqx:=X - q3*v[3][1,1];
```

$$line\_eqx := X - \frac{q3}{\alpha 3^3}$$

```
>line_eqy:=Y - q3*v[3][2,1];
```

$$line\_eqy := Y - \frac{q3}{\alpha 3^2}$$

```
>line_eqz:=Z - q3*v[3][3,1];
```

$$line\_eqz := Z - \frac{q3}{\alpha 3}$$

The system of linear equations which have to be solved is

```
>R1:=surf_eqx=line_eqx;
```

$$R1 := \frac{q1}{\alpha 1^3} + \frac{q2}{\alpha 2^3} = X - \frac{q3}{\alpha 3^3}$$

```
>R2:=surf_eqy=line_eqy;
```

$$R2 := \frac{q1}{\alpha 1^2} + \frac{q2}{\alpha 2^2} = Y - \frac{q3}{\alpha 3^2}$$

```
>R3:=surf_eqz=line_eqz;
```

$$R3 := \frac{q1}{\alpha 1} + \frac{q2}{\alpha 2} = Z - \frac{q3}{\alpha 3}$$

```
>SOL:=solve({R1,R2,R3},{q1,q2,q3}):
>Q1:=subs(SOL,q1);
```

$$Q1 := \frac{\alpha 1^3\,(-\alpha 2\,Y + X\,\alpha 2\,\alpha 3 + Z - \alpha 3\,Y)}{(\alpha 1 - \alpha 3)\,(-\alpha 2 + \alpha 1)}$$

```
>Q2:=subs(SOL,q2);
```

$$Q2 := -\frac{\alpha 2^3\,(-\alpha 1\,Y + \alpha 1\,X\,\alpha 3 + Z - \alpha 3\,Y)}{(-\alpha 3 + \alpha 2)\,(-\alpha 2 + \alpha 1)}$$

```
>Q3:=subs(SOL,q3);
```

$$Q3 := \frac{\alpha 3^3\,(-\alpha 2\,Y + \alpha 2\,\alpha 1\,X + Z - \alpha 1\,Y)}{-\alpha 2\,\alpha 3 - \alpha 1\,\alpha 3 + \alpha 3^2 + \alpha 1\,\alpha 2}$$

If q_1, q_2, q_3 fulfill $q_1 \in \langle U_j \quad U_{3-j}\rangle$, $q_2 \in \langle 0 \quad (U_{3-j} - U_j)\rangle$,
$q_3 \in \langle U_j - q_1 - q_2 \quad U_{3-j} - q_1 - q_2\rangle$ then the control signal in the linear segment is

```
>AZ:=collect(simplify(Q1+Q2+Q3),[X,Y,Z]);
```

$$AZ := \alpha 2\,\alpha 1\,X\,\alpha 3 + (-\alpha 1\,\alpha 3 - \alpha 1\,\alpha 2 - \alpha 2\,\alpha 3)\,Y + (\alpha 2 + \alpha 3 + \alpha 1)\,Z$$

which corresponds to the Ackerman's formula.

3 Synthesis of Algorithm in MAPLE for Segment $RS^1{}_1$

Let us consider the 3[rd] order integrator

$$\dot{\mathbf{x}} = \mathbf{A}\mathbf{x} + \mathbf{b}u \tag{4}$$

$$\begin{aligned}\dot{x} &= y\\ \dot{y} &= z\\ \dot{z} &= u\end{aligned}\ ;\quad \mathbf{x} = \begin{bmatrix} x\\ y\\ z\end{bmatrix}\ ;\quad \mathbf{b} = \begin{bmatrix} 0\\ 0\\ 1\end{bmatrix}\ ;\quad \mathbf{A} = \begin{bmatrix} 0 & 1 & 0\\ 0 & 0 & 1\\ 0 & 0 & 0\end{bmatrix}$$

The aim is to design and visualise a constrained pole assignment controller specified by the closed loop poles $\alpha_3 < \alpha_2 < \alpha_1 < 0$ and the control signal constraints

$$u \in \langle U_1, U_2 \rangle \tag{5}$$

that will drive the representative point $\mathbf{x}$ to a reference surface RS (invariant set of dimension 2) by decreasing the distance measured in the direction of the eigenvector

$$\mathbf{v}_i = [\alpha_i \mathbf{I} - \mathbf{A}]^{-1}\mathbf{b} = \begin{bmatrix} \dfrac{1}{\alpha_i^{\,3}} & \dfrac{1}{\alpha_i^{\,2}} & \dfrac{1}{\alpha_i} \end{bmatrix}^T \tag{6}$$

corresponding to $i = 3$. The RS itself is given by the two constrained modes of control corresponding to the first two poles.

After reaching RS, the controller should drive the reference point along RS to the reference curve RC (invariant set of dimension 1) , corresponding to the constrained mode of control associated with the pole α_1. Along this curve the transient will finally reach the origin that is an invariant set of dimension 0.

We start in the same way as in the linear case, at first we clear memory and read used packages, etc. The difference is that we have to use different parameters when solving equation (3) using $t_2 = 0, t_3 = 0, q_1 = U_1$ to get parameters t_1, q_2, q_3. However it is the same as to find the point of intersection of the surface (2) with the line defined by the vector $\mathbf{v}_3$ and the representative point. So the parametric representation of the surface mentioned above is

```
>surf_eqx:= subs({q1=U[1],t2=0},x12[1,1]);
```

$$surf\_eqx := U_1\left(\frac{1}{\alpha 1^3} - \frac{t1}{\alpha 1^2} + \frac{t1^2}{2\,\alpha 1}\right) - \frac{1}{6}U_1\, t1^3 + \frac{q2}{\alpha 2^3}$$

```
>surf_eqy:= subs({q1=U[1],t2=0},x12[2,1]);
```

$$surf\_eqy := U_1\left(\frac{1}{\alpha 1^2} - \frac{t1}{\alpha 1}\right) + \frac{1}{2}U_1\, t1^2 + \frac{q2}{\alpha 2^2}$$

```
>surf_eqz:= subs({q1=U[1],t2=0},x12[3,1]);
```

$$surf\_eqz := \frac{U_1}{\alpha 1} - U_1\, t1 + \frac{q2}{\alpha 2}$$

The line can be expressed in the same way as in the linear case

```
>line_eqx:=X - q3*v[3][1,1];
```

$$line\_eqx := X - \frac{q3}{\alpha 3^3}$$

```
>line_eqy:=Y - q3*v[3][2,1];
```

$$line\_eqy := Y - \frac{q3}{\alpha 3^2}$$

```
>line_eqz:=Z - q3*v[3][3,1];
```

$$line\_eqz := Z - \frac{q3}{\alpha 3}$$

Nevertheless, we have to solve the system of nonlinear equations in this case

```
>R1:=surf_eqx=line_eqx;
```

$$R1 := U_1\left(\frac{1}{\alpha 1^3} - \frac{t1}{\alpha 1^2} + \frac{t1^2}{2\,\alpha 1}\right) - \frac{1}{6}U_1\,t1^3 + \frac{q2}{\alpha 2^3} = X - \frac{q3}{\alpha 3^3}$$

```
>R2:=surf_eqy=line_eqy;
```

$$R2 := U_1\left(\frac{1}{\alpha 1^2} - \frac{t1}{\alpha 1}\right) + \frac{1}{2}U_1\,t1^2 + \frac{q2}{\alpha 2^2} = Y - \frac{q3}{\alpha 3^2}$$

```
>R3:=surf_eqz=line_eqz;
```

$$R3 := \frac{U_1}{\alpha 1} - U_1\,t1 + \frac{q2}{\alpha 2} = Z - \frac{q3}{\alpha 3}$$

Now we solve them. At first we solve the equation R3, for parameter q_2. The result is substituted in equation R2 and stored in R2s.

```
> Q2:=solve(R3,q2);
```

$$Q2 := \frac{(-U_1\,\alpha 3 + U_1\,t1\,\alpha 1\,\alpha 3 + Z\,\alpha 1\,\alpha 3 - q3\,\alpha 1)\,\alpha 2}{\alpha 1\,\alpha 3}$$

```
>R2s:=subs(q2=Q2,R2);
```

$R2s :=$

$$U_1\left(\frac{1}{\alpha 1^2} - \frac{t1}{\alpha 1}\right) + \frac{1}{2}U_1\,t1^2 + \frac{-U_1\,\alpha 3 + U_1\,t1\,\alpha 1\,\alpha 3 + Z\,\alpha 1\,\alpha 3 - q3\,\alpha 1}{\alpha 1\,\alpha 3\,\alpha 2} = Y - \frac{q3}{\alpha 3^2}$$

We solve R2s for t_1

```
>T1:=solve(R2s,t1);
```

$$T1 := \Big(-U_1\,\alpha 1\,\alpha 3 + \alpha 3\,U_1\,\alpha 2 + (U_1^{\,2}\,\alpha 1^2\,\alpha 3^2 - \alpha 3^2\,U_1^{\,2}\,\alpha 2^2 - 2\,U_1\,\alpha 2\,\alpha 1^2\,\alpha 3^2\,Z$$
$$+ 2\,U_1\,\alpha 2\,\alpha 1^2\,\alpha 3\,q3 - 2\,U_1\,\alpha 2^2\,q3\,\alpha 1^2 + 2\,U_1\,\alpha 2^2\,Y\alpha 1^2\,\alpha 3^2)^{(1/2)}\Big) \Big/ (U_1\,\alpha 2\,\alpha 3$$
$$\alpha 1), -\Big(U_1\,\alpha 1\,\alpha 3 - \alpha 3\,U_1\,\alpha 2 + (U_1^{\,2}\,\alpha 1^2\,\alpha 3^2 - \alpha 3^2\,U_1^{\,2}\,\alpha 2^2 - 2\,U_1\,\alpha 2\,\alpha 1^2\,\alpha 3^2\,Z$$
$$+ 2\,U_1\,\alpha 2\,\alpha 1^2\,\alpha 3\,q3 - 2\,U_1\,\alpha 2^2\,q3\,\alpha 1^2 + 2\,U_1\,\alpha 2^2\,Y\alpha 1^2\,\alpha 3^2)^{(1/2)}\Big) \Big/ (U_1\,\alpha 2\,\alpha 3$$
$$\alpha 1)$$

Then we get two results for t_1

$$\frac{-U_1\ \alpha 1\ \alpha 3 + \alpha 3\ U_1\ \alpha 2 + \sqrt{D}}{U_1\ \alpha 2\ \alpha 3\ \alpha 1}$$

$$-\frac{U_1\ \alpha 1\ \alpha 3 - \alpha 3\ U_1\ \alpha 2 + \sqrt{D}}{U_1\ \alpha 2\ \alpha 3\ \alpha 1}$$

Parameter t_1 has to be positive real, so there is easy to choose the first result because the second result is always negative. Now we substitute the results into the equation R1.

```
>tmp:=subs(q2=Q2,R1):R1s:=subs(t1=T1[1],tmp);
```

Then the parameter q_3 is the solution of R1s.

```
>Q3:=solve(R1s,q3);
```

After substitution $\alpha_1 = -0.5, \alpha_2 = -1$, $\alpha_3 = -2$, $U_1 = -1$, $U_2 = 1$ one gets

$$\begin{aligned} Q3subs := -(&\\ &-4.50\,(1.50\,Z + 4.50\,Y - 6.88 + 3.00\,X + \sqrt{D})^{(2/3)}\\ &+ 2.00\,(1.50\,Z + 4.50\,Y - 6.88 + 3.00\,X + \sqrt{D})^{(4/3)}\\ &+ 1.38 + 3.00\,Z + 9.00\,Y + 6.00\,X + 2.00\sqrt{D}\\ &- 5.50\,(1.50\,Z + 4.50\,Y - 6.88 + 3.00\,X + \sqrt{D})^{(1/3)}\,)\\ &\Big/\ (1.50\,Z + 4.50\,Y - 6.88 + 3.00\,X + \sqrt{D})^{(2/3)}\\ &- 4.00\,Y - 4.00\,Z \end{aligned}$$

$$\begin{aligned} D :=& -61.88\,Y + 27.00\,X\,Y + 13.50\,Y\,Z + 68.06\\ &- 41.25\,X + 20.25\,Y^2 - 20.63\,Z + 2.25\,Z^2 + 9.00\,X\,Z\\ &+ 9.00\,X^2 \end{aligned}$$

The result has to be real as well, so we have chosen the real result and using several substitutions we have got the control signal (AZ) in symbolic form.

```
>TT1:=simplify(subs(q3=Q3[1],T1[1])):
>QQ2:=simplify(subs({q3=Q3[1],t1=TT1},Q2)):
>AZ:=collect(simplify(Q3[1]+U[1]+QQ2),[X,Y,Z]);
```

The synthesis for $RS^j{}_{12}$ leads to the 6th order equation. The result is one of the roots of the 6th order polynomial, that can be chosen using additional conditions described in chapter 7 (i.e. $t_1 > 0, t_2 > 0$). The solution of (3), i.e. the synthesis of control algorithm, for $RS^j{}_2$ can be obtained in similar way as for the $RS^j{}_1$.

4 Extending Design Control to Other 3rd Order Plant

The possibility of controlling a different third order system using algorithm described above is presented in this chapter. We have chosen the plant with the system matrix $\mathbf{A}_X$ and the input vector $\mathbf{b}_X$

$$\mathbf{A}_X = \begin{bmatrix} 0 & 1 & 1 \\ 0 & 0 & 1 \\ 0 & 0 & 0 \end{bmatrix} \quad \mathbf{b}_X = \begin{bmatrix} 0 \\ 0 \\ 1 \end{bmatrix} \tag{7}$$

The following simulation (Fig. 1) shows the phase trajectory, output of the system and a control signal, where the closed loop poles have been $\alpha_1 = -0.1, \alpha_2 = -0.5, \alpha_3 = -1$, the control limits have been $U \in \langle -1 \;\; 1 \rangle$ and the initial condition has been $[1450, -130, 8]^{\mathrm{T}}$.



Figure 1: Output of the system, control signal, phase trajectory.

5 Conclusion

The constrained controller has been designed in computer algebra system MAPLE. Due to the symbolic solutions there is not difficult to change closed loop poles or the input constraints in any application of this control. The controller derived for the triple integrator can also be used for controlling a broader class of 3rd order systems.

6 References

Athans,M. and P.L.Falb (1966). Optimal Control, An Introduction to the Theory and its Applications. MacGraw-Hill, New York.

Bisták, P. Ťapák, P. and M. Huba (2006). *Constrained Pole Assignment Control of Double and Triple Integrator*, In: International journal of Computing Anticipatory Systems

Blanchini, F. and Miani, S. (1997). Nonlinear controllers for the constrained stabilization of uncertain dynamic systems" in *Control of constrained systems*, Edited by S. Tarbourich and J.C. Hennet, Springer Verlag.

Borelli, Francesco (2003). *Constrained optimal control of linear and hybrid systems*. Berlin: Springer.

Feldbaum A.A., *Optimal control systems*, Academic Press, N.York. (1965).

Glattfelder,A.H. und Schaufelberger,W. (2003). *Control Systems with Input and Output Constraints*. Springer London.

Goodwin, G. C., Seron, M. M., De Doná, J. A. (2005). *Constrained Control and Estimation, An Optimisation Approach*. Springer London.

Hu, T. and Z. Lin (2001). Control systems with actuator saturation: Analysis and design. Birkhäuser, Boston, MA.

Huba, M. (2003). *Constrained systems design. Vol.1 Basic controllers*. STU Bratislava (in Slovak).

Huba, M. (2005). Constrained Pole Assignment Control – Real & Complex Poles. CASYS 2005, Liege.

Liu, D. and A. N. Michel (1994). Dynamical Systems With Saturation Nonlinearities: Analysis and Design. New York: Springer-Verlag.

Pavlov, A., A. (1966). Synthesis of relay time-optimal systems, in Russian, Publishing house Nauka Moscow.

Perez, T. (2005). Ship motion control: course keeping and roll stabilisation using rudder and fins. Springer London 2005.

Saberi, A., Stoorvogel, A. and P. Sannuti, (1999). *Output Regulation and Control Problems with Regulation Constraints*. New York: Springer-Verlag.

Saberi, A., Stoorvogel, A. and P. Sannuti, (2000). *Control of linear systems with regulation and input constraints*, Springer Verlag, London.

Smith O.J.M. (1958) *Feedback control systems*, McGraw-Hill.

Tabouriech, S. and G. Garcia, editors, (1997). Control of Uncertain Systems with Bounded Inputs, volume 227 of Lecture Notes in Control and Information Science. Springer Verlag, London.